
Initiation à la programmation sur VBA 
  

Feuille 8 

 

Les tableaux, le type Range et le type Variant pour le VBA 

 

1. Exercices 

Le type « Range » 

Exercice 1. La plus petite valeur dans une plage 

Créez une fonction qui renvoie la plus petite valeur dans une plage de cellule.  

 

Indice : Prendre en repère la première cellule. Parcourir les cellules, si le contenu de la cellule courante 

est plus petit que le repère, il devient le nouveau repère. Attention, vous devez utiliser la boucle For 

Each. 

'Chercher un minimum dans une plage de cellules (avec For Each) 

Public Function MonMin(plage As Range) As Double 

'cellule à déplacer dans la plage 

Dim cellule As Range 

Dim repere As Double 

 

'repere 

repere = plage.Cells(1, 1).Value 

 

'parcourir 

For Each cellule In plage 

     If (cellule.Value < repere) Then 

        repere = cellule.Value 

     End If 

    Next cellule 

 

   'le repère est aussi la valeur minimale 

   MonMin = repere 

End Function 

Exercice 2. La plus grande valeur de la plage 

Créez une fonction qui renvoie la plus grande valeur dans une plage de cellule.  

 

Indice : Même schéma que minimum, mais en inversant le sens des comparaisons. 

 

'Chercher un maximum dans une plage de cellules (avec For Each) 

Public Function MonMax(plage As Range) As Double 

'cellule à déplacer dans la plage 

Dim cellule As Range 

Dim repere As Double 

'repere 

repere = plage.Cells(1, 1).Value 

'parcourir 

For Each cellule In plage 

    If (cellule.Value > repere) Then 

        repere = cellule.Value 

    End If 

Next cellule 

'le repère est aussi la valeur maximale 

MonMax = repere 

End Function 

Boucle « For Each » 

Exercice 3. Moyenne robuste 

Cette fonction doit calculer la moyenne des valeurs d’une plage de cellules, après avoir retranché la plus 

petite et la plus grande valeur. 

 

Exemple : Dans la plage  la fonction doit calculer la moyenne des 

valeurs (14, 15, 13, 18, 16, 17, 12), en excluant donc 2 et 95. Attention, si le nombre de cellules 

de la plage est inférieur ou égal à 2, on renvoie la chaîne de caractères « calcul impossible ». 

 

'Moyenne des valeurs en enlevant le min et le max 

Public Function MaMoyenneRobuste(plage As Range) As Variant 

'compter les cellules 

Dim nbCell As Double 

'somme totale 

Dim somme As Double 

'cellule 

Dim cellule As Range 

'nombre de cellules 

nbCell = plage.Rows.Count * plage.Columns.Count 

'ou bien >> nbCell = plage.Cells.Count << est valable aussi 

'tester s'il y en a suffisamment (de cellules) 

If (nbCell > 2) Then 

    'calculer la somme 

    somme = 0 

    For Each cellule In plage 

        somme = somme + cellule.Value 

    Next cellule 

    'retrancher de somme le min et le max 

    somme = somme - (MonMin(plage) + MonMax(plage)) 

    MaMoyenneRobuste = somme / (nbCell - 2) 

Else 

    'renvoyer une valeur indiquant l'erreur 

    MaMoyenneRobuste = "calcul impossible" 

End If 

'fin de fonction 

End Function 

 

Exercice 4. Moyenne non vide 

Créez une fonction VBA qui prend en entrée une plage de cellules. Elle doit calculer la moyenne des 

valeurs en prenant en compte uniquement les cellules non vides. Attention, dans le cas où toutes les 

cellules sont vides, la fonction doit renvoyer la chaîne « pas de cellules valides ».  

 

Remarque : La fonction IsEmpty() permet de savoir si une cellule est vide (elle renvoie TRUE) ou 

non (elle renvoie FALSE). 

 

Public Function MaMoyenneNonVide(plage As Range) As Variant 

Dim somme As Double 'pour sommer 

Dim cellule As Range 'pour parcourir les cellules de la plage 

Dim nbCellules As Long 'compteur de cellules valides 

Dim resultat As Variant 

 

nbCellules = 0 'initialisation du compteur 

somme = 0 'initialisation de la somme 

 

'parcourir toutes les cellules 

For Each cellule In plage 

    If (IsEmpty(cellule) = False) Then 

        nbCellules = nbCellules + 1 

        somme = somme + cellule.Value 

    End If 

Next cellule 

 

'toutes les cellules ont-elles été vides ? 

If (nbCellules = 0) Then 

    resultat = "pas de cellules valides" 

Else 

    resultat = somme / nbCellules 

End If 

 

'renvoyer les résultats 

MaMoyenneNonVide = resultat 

 

End Function 

Les tableaux 

Exercice 5. Valeurs cumulées 

Créez une procédure val_cumulees() qui contient : 

• un premier tableau tab1 contient les dix valeurs numériques données par l’utilisateur ; 

• un second tableau tab2 enregistre les valeurs cumulées. 
La ième valeur cumulée est calculée en faisant la somme de la ième valeur du tableau initial 
tab1(i) et de la valeur cumulée tab2(i – 1). 
 

Sub val_cumulees() 

Dim tab1(1 To 10) As Integer, tab2(1 To 10) As Integer 

Dim i As Integer 

Dim ch1 As String, ch2 As String 

ch1 = "" 

For i = 1 To 10 

tab1(i) = InputBox("donne la valeur n°" & i) 

ch1 = ch1 & CStr(tab1(i)) & " " 

Next i 

tab2(1) = tab1(1) 

ch2 = tab2(1) & " " 

For i = 2 To 10 

tab2(i) = tab1(i) + tab2(i - 1) 

ch2 = ch2 & CStr(tab2(i)) & " " 

Next i 

MsgBox ch1 & vbCr & ch2 

End Sub 

Exercice 6. Appel sommation 

Créez une procédure appel-sommation() qui crée et initialise un tableau en demandant à 

l’utilisateur la taille du tableau à créer et les différentes valeurs à y ranger. Cet algorithme fera 
ensuite appel à la fonction sommation et affichera son résultat. 

 
Public Function sommation(t() As Integer) As Integer 

Dim s As Integer, i As Integer 

s = 0 

For i = LBound(t) To UBound(t) 

s = s + t(i) 

Next 

sommation = s 

End Function 

 

 

Public Sub appel_sommation() 

Dim t() As Integer, n As Integer, i As Integer 

n = InputBox("Taille du tableau ?") 

ReDim t(1 To n) 

For i = 1 To n 

t(i) = InputBox("Valeur t(" & i & ") ?") 

Next 

MsgBox ("Somme des éléments du tableau = " & sommation(t)) 

End Sub 

 

Exercice 7. Déclarer et utiliser un tableau à deux dimensions 

Déclarez et utilisez un tableau multidimensionnel de type entier comportant 5 lignes et 5 colonnes (option 

de base par défaut). Ce tableau est local à une procédure nommée Tableau2D(). Nommez-le 

intTab2D. Initialisez ce tableau avec des nombres aléatoires de 1 à 100. Affichez-le en effectuant un 

parcours ligne colonne. 

 

Exemple :  

 

 

 

 

 

 

 

 

 

 

 

 

Affichez la fenêtre 

Exécution comme 

ceci :  

 

 

 

 

 

 

 

 

 

 

 

Option Base 0 

 

Sub Tableau2D() 

    Dim intTab2D(1 To 5, 1 To 5) As Integer 

    Dim intLigne As Integer 

    Dim intColonne As Integer 

    VBA.Randomize 

    For intLigne = 1 To 5 

        For intColonne = 1 To 5 

            intTab2D(intLigne, intColonne) = (100 * VBA.Rnd) + 1 

            Debug.Print "Ligne", intLigne, "Colonne", 

intColonne, "-->", Format(intTab2D(intLigne, intColonne), "@@@") 

        Next intColonne 

    Next intLigne 

End Sub 

 

Exercice 8. Déclarer et utiliser un tableau dynamique 

Dans une procédure TabDynamique(), créez un tableau dynamique, de niveau module, à une colonne 

comportant des nombres. Nommez ce tableau dblTabDynamique. Proposez une boite de saisie pour 

affecter au dernier élément la taille du tableau. Affichez cet élément. 

 

 

 

 

 

Commentaire : La taille maximale d’un tableau varie en fonction de votre système d’exploitation et 

de la mémoire disponible. L’utilisation d’un tableau dépassant la quantité de RAM disponible ralentit le 

traitement. VBA peut aussi arrêter l’exécution de la procédure. 

 

ReDim permet de modifier la taille d’un tableau dynamique aussi souvent que nécessaire. Attention : 

ReDim ne peut être utilisé que dans une procédure. 

 

Option Base 0 

 

Private dblTabDynamique() As Double 

 

Sub TabDynamique() 

    Dim intTaille As Integer 

    On Error GoTo SortieErreur 

    intTaille = Application.InputBox(prompt:="Indiquez la taille 

du tableau " & "(entre 1 et 10000)", Type:=1) 

    intTaille = VBA.CInt(intTaille) 

    ReDim dblTabDynamique(1 To intTaille) 

    dblTabDynamique(intTaille) = intTaille 

    Debug.Print "dernier élément avec taille du tableau :"; 

dblTabDynamique(intTaille) 

SortieErreur: 

End Sub 

2. Exercices optionnels 

Exercice 9. Initialiser un tableau avec une boucle 

Dans une procédure TableauBoucle, remplissez un tableau local nommé intTabInit avec 10 

nombres entiers aléatoires de 1 à 100. Affichez tous les nombres avec le total du tableau. 

 

 
 

Indice : utilisez l’instruction Randomize qui initialise le générateur de nombres aléatoires et la fonction 

Rnd.  

Exemple : 

Dim intMyValue As Integer 

Randomize ‘Initialise le générateur de nombres aléatoires. 

‘ Renvoie une valeur aléatoire comprise entre 1 et 6. 

intMyValue = Int((6*Rnd)+1) 

Utilisez le boucle For … Next pour remplir le tableau. 

 

Sub TableauBoucle() 

    Dim intCompteur As Integer 

    Dim intTotal As Integer 

    Dim intTabInit(1 To 10) As Integer 

    VBA.Randomize 

    For intCompteur = 1 To 10 

        intTabInit(intCompteur) = (100 * VBA.Rnd) + 1 

        Debug.Print "Nombre"; Format(intCompteur, "@@@"); ":"; 

Format(intTabInit(intCompteur), "@@@") 

        intTotal = intTotal + intTabInit(intCompteur) 

    Next intCompteur 

    Debug.Print "Total", intTotal 

End Sub 

 

(100 * VBA.Rnd) + 1 : pour éviter d’avoir le zéro parmi les nombres aléatoires. La 

fonction Format permet de mettre en forme les résultats. Elle n’est pas indispensable au 

bon fonctionnement de la procédure. 

Exercice 10. Déclarer et utiliser un tableau à plus de deux dimensions 

Déclarez un tableau tridimensionnel de type entier, de niveau module, de 10 par 10 par 50. Nommez-le 

intTab3D. 

 

Créez une procédure nommée procTab3D. Affectez au 5000ème élément du tableau multidimensionnel 

intTab3D la valeur 777. Affichez cet élément. 

 

 
 

Déclaration du tableau dans la partie Déclarations du module 

Private intTab3D(1 To 10, 1 To 10, 1 To 50) As Integer 

 

Sub procTab3D() 

    intTab3D(10, 10, 50) = 777 

    Debug.Print "5000ème élément :"; intTab3D(10, 10, 50) 

End Sub 

 

Exercice 11. Agrandir un tableau dynamique en préservant les valeurs 

initiales 

Dans une procédure agrandirTabDynamique, utilisez le tableau dynamique dblTabDynamique 

(voir exercice 8). Demandez une première taille puis une seconde taille pour agrandir le tableau. 

Affichez la première taille et la dernière taille choisies en se référant aux éléments 

correspondants du tableau. 

 

Exemple avec dblTabDynamique(intTaille) = 7 et 

dblTabDynamique(intTaille) = 9 : 

 

 

 

 

 

 

 

 

 

 

Indice : Utilisez le mot-clé Preserve afin de conserver les premiers éléments du tableau. 

 

Sub agrandirTabDynamique() 

    Dim intTaille1 As Integer 

    Dim intTaille2 As Integer 

     

    intTaille1 = Application.InputBox(prompt:="Première taille 

du tableau :", Type:=1, Title:=" Nombre entier positif < 32767") 

    ReDim dblTabDynamique(1 To intTaille1) 

    dblTabDynamique(intTaille1) = intTaille1 

    Debug.Print "Première taille"; dblTabDynamique(intTaille1) 

    intTaille2 = Application.InputBox(prompt:="Deuxième taille 

du tableau :", Type:=1, Title:=" Nombre entier positif < 32767") 

    ReDim Preserve dblTabDynamique(1 To intTaille2) 

    dblTabDynamique(intTaille2) = intTaille2 

    Debug.Print "Deuxième taille"; dblTabDynamique(intTaille2) 

End Sub 

 

Exercice 12. Exploiter un tableau à l’aide d’une boucle 

Créez une procédure ExploiterTableau qui affiche une table de multiplication selon le choix de 

l’utilisateur. 

Exemple : 

 

 

Indice : Cet exemple utilise la **CDbl ** fonction pour convertir une expression en Double. 

Dim MyCurr, MyDouble  

MyCurr = CCur(234.456784) ' MyCurr is a Currency.  

MyDouble = CDbl(MyCurr * 8.2 * 0.01) ' Convert result to a Double. 

 

 

Sub ExploiterTableau() 

    Dim dblNombre As Integer 

    Dim intCompteur As Integer 

    Dim dblTableMultiplication(1 To 10) As Double 

    dblNombre = Application.InputBox(prompt:="Entrez le nombre 

pour la table" & " de multiplication.", Type:=1) 

    If dblNombre <> 0 Then 

        dblNombre = VBA.CDbl(dblNombre) 

        For intCompteur = 1 To 10 

            dblTableMultiplication(intCompteur) = intCompteur * 

dblNombre 

            Debug.Print dblNombre, "x", VBA.Format(intCompteur, 

"@@"), "=", Format(dblTableMultiplication(intCompteur), "@@@@") 

        Next intCompteur 

    End If 

End Sub 

